// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/Buy.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.9;

import "@openzeppelin/contracts/access/Ownable.sol";
import "./Public.sol";

struct Item {
    string name;
    uint128 price;
    uint64 txLimit;
    uint64 walletLimit;
    uint64 supplyLimit;
}

contract Buy is Ownable {
    IERC20 public token;
    bool public enabled;

    address private _cashier;
    mapping(uint16 => Item) private _items;
    mapping(uint16 => mapping(address => uint256)) private _purchased;

    event Purchase(
        address indexed purchaser,
        uint16[] items,
        uint16[] quantities,
        uint256 total,
        uint256 timestamp
    );

    constructor(address tokenAddress) {
        token = IERC20(tokenAddress);
    }

    function setEnabled(bool b) public onlyOwner {
        enabled = b;
    }

    function setCashier(address a) public onlyOwner {
        _cashier = a;
    }

    function setItem(
        uint16 id,
        string memory name,
        uint128 price,
        uint64 txLimit,
        uint64 walletLimit,
        uint64 supplyLimit
    ) public onlyOwner {
        _items[id] = Item({
            name: name,
            price: price,
            txLimit: txLimit,
            walletLimit: walletLimit,
            supplyLimit: supplyLimit
        });
    }

    function deleteItem(uint16 id) public onlyOwner {
        delete _items[id];
    }

    function buy(uint16 item, uint16 quantity) external {
        require(enabled == true, "Purchasing is disabled.");
        require(_cashier != address(0), "Cashier not configured.");

        require(_items[item].price > 0, "Item not configured.");
        require(
            _items[item].txLimit == 0 || quantity <= _items[item].txLimit,
            "Quantity exceeds transaction limit."
        );
        require(
            _items[item].walletLimit == 0 ||
                quantity + _purchased[item][_msgSender()] <=
                _items[item].walletLimit,
            "Quantity exceeds wallet limit."
        );
        require(
            _items[item].supplyLimit == 0 ||
                quantity + _purchased[item][address(0)] <=
                _items[item].supplyLimit,
            "Quantity exceeds supply limit."
        );

        uint256 total = quantity * _items[item].price;

        require(token.balanceOf(_msgSender()) >= total, "Insufficient funds.");

        require(
            token.allowance(_msgSender(), address(this)) >= total,
            "Not approved."
        );

        token.transferFrom(_msgSender(), _cashier, total);

        if (_items[item].walletLimit > 0) {
            _purchased[item][_msgSender()] += quantity;
        }

        if (_items[item].supplyLimit > 0) {
            _purchased[item][address(0)] += quantity;
        }

        emit Purchase({
            purchaser: _msgSender(),
            items: uint16singleton(item),
            quantities: uint16singleton(quantity),
            total: total,
            timestamp: block.timestamp
        });
    }

    function purchased(
        uint16 item,
        address addr
    ) external view returns (uint256) {
        return _purchased[item][addr];
    }

    function getItem(uint16 item) external view returns (Item memory) {
        return _items[item];
    }

    function uint16singleton(
        uint16 u
    ) internal pure returns (uint16[] memory a) {
        a = new uint16[](1);
        a[0] = u;
        return a;
    }
}

// File: contracts/Public.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(
        address indexed owner,
        address indexed spender,
        uint256 value
    );

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(
        address owner,
        address spender
    ) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

interface IBuy {
    struct Item {
        string name;
        uint128 price;
        uint64 txLimit;
        uint64 walletLimit;
        uint64 supplyLimit;
    }

    event Purchase(
        address indexed purchaser,
        uint16[] items,
        uint16[] quantities,
        uint256 total,
        uint256 timestamp
    );

    function enabled() external view returns (bool);

    function buy(uint16 item, uint16 quantity) external;

    function purchased(
        uint16 item,
        address addr
    ) external view returns (uint256);

    function getItem(uint16 item) external view returns (Item memory);
}